

Percobaan

5

JUDUL :

TIMER/COUNTER
Menggunakan DT-51 MinSys

TUJUAN :

- Menggunakan Timer/Counter pada DT-51 Minimum System sebagai timer ataupun sebagai counter.
- Memanfaatkan Special Function Register (SFR) untuk mengatur Timer/Counter yang terdapat pada Timer Mode (TMOD) dan Timer Control (TCON).
- Membuat program menggunakan Timer/Counter dalam mode 0, 1, dan 2 beserta pin T0 dan T1 dari DT-51 Minimum System.

TEORI :

MCS-51 telah menyediakan dua buah Timer/Counter yang dapat digunakan sebagai timer ataupun sebagai counter. Masing-masing Timer/Counter memiliki empat mode dan dapat dikombinasikan.

Konsep Timer dan Counter


Pada dasarnya timer dan counter merupakan system yang sama-sama menambahkan diri hingga overflow. Timer memanfaatkan frekuensi osilator untuk bertambah tiap machine cycle. Counter memanfaatkan sumber dari luar IC MCS-51. Counter akan bertambah jika terdapat transisi dari nilai "1" ke nilai "0" pada pin T0 (P3.4) atau T1 (P3.5). Transisi counter sama halnya falling edge trigger pada external

interrupt dimana nilai "1" harus valid selama 1 machine cycle lalu disusul nilai "0" yang valid selama 1 machine cycle.

Timer dapat digunakan untuk menghitung suatu periode waktu antara kejadian, sebagai jarak waktu antara kejadian, dan menghasilkan baud rate untuk komunikasi serial. Counter dapat digunakan untuk menghitung jumlah munculnya suatu kejadian. Misalkan terdapat sebuah ban berjalan yang memasukkan sejumlah permen ke dalam masing-masing kantong plastik. Jika diinginkan sebuah kantong diisi dengan permen sebanyak 100 buah, maka sistem tersebut dapat menggunakan counter untuk menghitung banyaknya permen. Jika diinginkan sebuah kantong diisi dengan permen selama dua detik tanpa memperdulikan jumlah permen yang diisikan ke dalam kantong, maka sistem tersebut dapat menggunakan timer untuk menghitung waktu pengisian kantong.

Timer Register

Register yang digunakan untuk mengatur timer/counter terdapat pada Timer Mode (TMOD) dan Timer Control (TCON).



Gambar 13.1. Alokasi Bit TMOD

Tabel 13-1. Timer Mode

Simbol	Deskripsi
GATE	Pemilih external atau internal control
C/T	Pemilih Timer atau Counter
M1	Pemilih Mode Timer / Counter
M0	Pemilih Mode Timer / Counter

Penjelasan dari masing-masing bit TMOD pada tabel 6-1 adalah sebagai berikut:

- GATE

Jika GATE dan TRx (TR0 atau TR1 pada TCON) diberi nilai "1" oleh user, maka Timer/Counter hanya beroperasi jika pin \overline{INTx} bernilai high.

Jika GATE bernilai "0", maka Timer/Counter hanya beroperasi jika TRx bernilai "1".

- C/\overline{T}

C/\overline{T} harus diberi nilai "1" oleh program user untuk menjalankan mode counter dan diberi nilai "0" untuk menjalankan mode timer.

- M1 dan M0

M1 dan M0 merupakan dua bit pemilih mode operasi timer/counter.

	MSB		LSB
TF1	TR1	TF0	TR0
IE1	IT1	IE0	IT0

Gambar 13.2. Alokasi Bit TCON

Tabel 13-2. Timer Control

Bit	Alamat Bit	Simbol	Deskripsi
TCON.7	8FH	TF1	Timer/Counter 1 Overflow Flag
TCON.6	8EH	TR1	Timer 1 Run Control Bit
TCON.5	8DH	TF0	Timer/Counter 0 Overflow Flag
TCON.4	9CH	TR0	Timer 0 Run Control Bit
TCON.3	8BH	IE1	External Interrupt 1 Edge Flag
TCON.2	8AH	IT1	External Interrupt 1 Type Control Bit
TCON.1	89H	IE0	External Interrupt 0 Edge Flag
TCON.0	88H	IT0	External Interrupt 1 Type Control Bit

Penjelasan masing-masing bit TCON yang berkaitan dengan timer/counter sebagai berikut:

- TF1

TF1 diberi nilai "1" secara hardware saat nilai timer/counter 1 (TH1 dan TL1) mengalami overflow. TF1 akan diberi nilai "0" secara hardware saat mikrokontroler melompat ke Interrupt Service Routine.

- TR1
TR1 harus diberi nilai "1" oleh program user untuk menjalankan timer 1.
- TF0
TF0 diberi nilai "0" secara hardware saat nilai timer/counter 0 (TH) dan TL0) mengalami overflow. TF0 akan diberi nilai "0" secara hardware saat mikrokontroler melompat ke Interrupt Service Routine.
- TR0
TR0 harus diberi nilai "1" oleh program user untuk menjalankan timer 0.

Mode Operasi

Timer/Counter memiliki 4 mode operasi seperti ditunjukkan pada tabel 6-3. Setiap mode memiliki karakteristik tersendiri. User harus benar-benar memahami masing-masing mode agar dapat memilih mode yang tepat dalam program yang dibuat.


Tabel 13-3. Mode Operasi Timer/Counter

M1	M0	Mode Operasi	
0	0	0	Timer/Counter, 13 bit
0	1	1	Timer/Counter, 16 bit
1	0	2	Timer/Counter, 8 bit Auto Reload
1	1	3	Split Timer Mode untuk Timer/Counter
1	1	3	Timer/Counter 1 berhenti

❖ Mode 0

Mode 0 adalah timer/counter 13 bit. Mode ini disediakan terutama untuk menjaga kompatibilitas dengan prosesor 8048. Pada mode 0, register TLx (TL0 dan TL1) hanya digunakan 5 bit terendah saja sedangkan register THx (TH0 dan TH1) tetap selebar 8 bit.

TLx akan bertambah hingga bernilai 1Fh. Saat ada perubahan nilai TLx dari 1Fh ke 00h, THx akan bertambah 1. Nilai maksimal THx dan TLx adalah FF1Fh (THx = FFh dan TLx = 1Fh). Overflow akan terjadi jika ada perubahan dari FF1Fh ke 0000h. Nilai THx dan TLx dapat diubah oleh user setiap saat dalam program. Gambar timer/counter mode 0 diperlihatkan sebagai berikut :



Gambar 13.3. Timer/Counter Mode 0

❖ Mode 1

Mode 1 pada dasarnya serupa dengan mode 0. Namun pada mode 1 bit TLx digunakan sehingga mode 1 merupakan timer/counter 16 bit. TLx akan bertambah hingga bernilai FFh. Saat ada perubahan nilai TLx dari FFh ke 00h, THx akan bertambah 1. Nilai maksimal THx dan TLx adalah FFFFh (THx = FFh dan TLx = FFh). Overflow akan terjadi jika ada perubahan dari FFFFh ke 0000h. Nilai THx dan TLx dapat diubah oleh user setiap saat dalam program.

❖ Mode 2

Mode 2 adalah timer/counter 8 bit dengan fasilitas auto reload. TLx bertindak sebagai timer/counter 8 bit. Sedangkan THx berisi suatu nilai tertentu. Auto reload adalah fasilitas dimana nilai TLx setelah overflow tidak kembali ke 00h namun nilai TLx akan diambil dari nilai THx.

Misalkan THx berisi 47h dan TLx berisi FFh. Jika ada overflow pada TLx, Nilai TLx akan berubah dari FFh ke 47h, sesuai dengan nilai THx. Proses tersebut dikerjakan secara otomatis secara hardware sehingga user tidak perlu memeriksa apakah nilai TLx sudah overflow dan tidak perlu mengisi ulang dengan nilai yang baru. Mode 2 ini juga digunakan untuk menghasilkan baudrate.

❖ Mode 3

Pada mode 3, timer/counter 0 akan menjadi dua timer/counter 8 bit, sedangkan timer/counter 1 akan berhenti. TLO akan menjadi timer/counter 8 bit yang dikendalikan oleh bit kontrol timer/counter 0. TH0 akan menjadi timer 8 bit (bukan counter) yang dikendalikan oleh bit kontrol timer/counter 1 (meliputi TR1 dan TF1). Jadi TR1 mengendalikan TH0 dan timer/counter 1 secara bersamaan. Timer/counter 1 masih dapat dioperasikan dalam mode selain mode 3. Umumnya timer/counter 1 dioperasikan dalam mode 2 dan berfungsi sebagai penghasil baudrate. Jika user ingin menghentikan timer/counter 1 tanpa menghentikan TH0, user dapat memindahkan timer/counter 1 ke dalam mode 3.

INISIALISASI SEBUAH TIMER

Proses inisialisasi adalah proses menentukan nilai semua register yang berkaitan dengan timer/counter yang akan digunakan agar timer/counter dapat berfungsi sebagaimana yang dikehendaki.

Register yang harus diatur terlebih dahulu meliputi:

1. TMOD dan TCON

Langkah pertama adalah menentukan mode yang akan digunakan, fungsi yang dipilih (sebagai timer/counter), dan jenis kontrol (external control melalui pin INTx atau internal control melalui TRx).

Pengaturan TCON dilakukan untuk menjalankan timer.

Misalnya timer/counter 0 digunakan sebagai timer dalam mode 2 dengan external control dan timer/counter 1 digunakan sebagai counter dalam mode 1 dengan internal control, maka instruksi adalah sebagai berikut:

```
MOV    TMOD, #01011010b
atau
MOV    TMOD, #5Ah
```

Sedangkan untuk menjalankan kedua timer, instruksinya adalah sebagai berikut:

```
MOV    TCON, #01010000b
atau
MOV    TCON, #50h
atau
SETB   TR0
SETB   TR1
```

2. THx dan/atau TLx

Jika diperlukan, inisialisasi atau perubahan terhadap nilai THx dan/atau TLx dapat dilakukan dengan cara mengisi nilai tertentu ke dalam register tersebut. Misalnya register timer/counter 0 diisi dengan nilai 814Ah dan register timer/counter 1 diisi dengan nilai 0CF32h, maka instruksinya adalah sebagai berikut:

```
MOV    TH0 , #81H
MOV    TL0 , #4Ah
MOV    TH1 , #0CFh
MOV    TL1 , #32h
```

3. IE dan/atau IP

Jika timer/counter yang diprogram akan digunakan sebagai sumber interrupt, maka IE dan/atau IP juga harus diatur. Misalnya kedua timer/counter digunakan sebagai sumber interrupt dengan prioritas timer/counter 1, maka instruksinya adalah sebagai berikut:

```
MOV    IP , #00001000b
MOV    IE , #10001010b
atau
MOV    IP , #08h
MOV    IE , #8Ah
atau
SETB   PT1
SETB   ET1
SETB   ET0
SETB   EA
```

PERALATAN :

- Modul DT-51
- PC beserta software
 - Debugger
 - Downloader
 - Editor
- LED sebagai piranti output
- Switch toggle sebagai piranti input

PROSEDUR :

1. Hardware

- Hubungkan konektor (kabel pita) Port 1, Port C pada DT-51 ke *Modul Display LED* jika output yang digunakan adalah Port 1 atau Port C. Tetapi bila outputnya adalah Port A atau Port B maka hubungkan konektor (kabel pita) Port A atau Port B pada DT-51 ke *Modul Display LED* (seperti pada gambar 5-5).


Selanjutnya

- Hubungkan kabel serial dari konektor DB-9 pada DT-51 ke PC pada Port serial.
- Hubungkan catu daya 9 Volt AC pada modul DT-51.
- Pin Konektor Select pada posisi Pin 1 dan 2 untuk mode *download program*, pada posisi Pin 2 dan 3 untuk mode *stand alone* , setelah download program berhasil.

2. Software

Editor :

- Tulislah program pada editor DOS prompt, Notepad atau lainnya.
- Program / file baru tersebut, simpan dalam direktori(folder) DT51 dengan extension *[namafile].ASM*.
- Exit



Gambar 13.4 Koneksi DT-51 MinSys dengan Display LED (akses memori)

Debugger :

- Compile file baru dengan perintah `C:\DT51\ASM51 [nama file].ASM` pada DOS Prompt, dengan terlebih dahulu pindah ke direktori DT51.
- Maka akan terbentuk file-file yang berekstension .HEX, .OBJ, dan .LST disamping itu juga muncul pesan, bahwa file yang telah dicompile tersebut terdapat kesalahan atau tidak.
- Bila ada kesalahan, bisa dilihat letak kesalahannya dengan membuka file yang berekstension .LST, melalui `C:\DT51\edit`.
- Untuk memperbaiki kesalahan (error), buka file yang berekstension .ASM. Jangan lupa disimpan kemudian dicompile kembali.
- Exit

Downloader :

- Untuk mendownload program yang telah sukses dicompile, maka gunakan perintah `C:\DT51\DT51L [nama file].HEX`
- Apabila proses download selesai, maka akan muncul pesan Download Succeeded.
- Bila gagal, maka akan muncul beberapa prosedur untuk mengecek kembali kesalahan tersebut.

PROGRAM PERCOBAAN :

- *Program 1. Program Timer/counter 0 sebagai Timer mode 0/1 untuk LED bit 5 pada Port 1 berkedip.*

```

$MOD51

                                CSEG
                                ORG 4000H
                                LJMP Start

                                ORG 400BH
COUNT:                         INC R0
                                CJNE R0,#10H,OUT1
                                SETB P1.5
                                MOV R7,#0FFH
LUP:                             MOV R6,#0FFH
                                DJNZ R6,$
                                DJNZ R7,LUP
                                CLR P1.5
                                MOV R0,#00H
OUT1:                            MOV TH0,#2CH
                                MOV TL0,#00H
```

```

                                RETI

;inisialisasi
                                ORG 4200H
Start:                          MOV SP,#30H
                                MOV R0,#00H
                                MOV TMOD,#01H
                                MOV TH0,#2CH
                                MOV TL0,#00H
                                MOV TCON,#10H
                                MOV P1,#00H
                                MOV IE,#82H
                                SJMP $
                                END

```

- **Program 2. Program Timer/counter 1 sebagai Timer mode 2 untuk nyala LED pada Port 1 bergantian, jika ada interrupt LED bit 5 nyala.**

```

$MOD51

                                CSEG
                                ORG 4000H
                                LJMP Start

                                ORG 401BH
COUNT:                          INC R1
                                CJNE R1,#0FFH,OUT1
                                INC R0
                                CJNE R0,#44H,OUT1
                                MOV P1,#00H
                                SETB P1.5
                                LCALL SDELAY
                                LCALL SDELAY
                                MOV R1,#00H
                                MOV R0,#00H
OUT1:                             RETI
                                ORG 4100H
SDELAY:                          PUSH 7
                                PUSH 6
                                MOV R7,#0FFH
LUP:                             MOV R6,#0FFH
                                DJNZ R6,$
                                DJNZ R7,LUP
                                POP 6
                                POP 7
                                RET
DELAY:                           MOV R7,#04H
LUP1:                            MOV R6,#0FFH
LUP2:                            MOV R5,#0FFH
                                DJNZ R5,$
                                DJNZ R6,LUP2
                                DJNZ R7,LUP1
                                RET

```

```

;inisialisasi
Start:    MOV    SP,#30H
          MOV    R1,#00H
          MOV    R0,#00H
          MOV    TMOD,#20H
          MOV    TH1,#01H
          MOV    TL1,#02H
          SETB   TR1
          MOV    IE,#88H

;main program
LOOP1:   MOV    P1,#00110011B
          LCALL  DELAY
          MOV    P1,#11001100B
          LCALL  DELAY
          SJMP  LOOP1
          END

```

TUGAS :

1. Buatlah program untuk menampilkan LED di port B dengan syarat:
 - semua LED dimulai dari keadaan diam
 - gunakan timer/counter 0 sebagai timer mode 0, dimulai dari '0000h'.
 - setiap kali terjadi interrupt, LED akan menampilkan bilangan biner yang bertambah mulai dari 00h, 01h, 02h, ..., 0ffh kembali lagi ke 00h.
 - Masing-masing interrupt hanya akan menampilkan satu nilai (interrupt pertama menampilkan '00000000b', interrupt kedua menampilkan '00000001b', dst)
2. Buatlah program untuk menampilkan LED di port B dengan syarat:
 - semua LED dimulai dari keadaan diam
 - gunakan timer/counter 1 sebagai timer mode 2
 - jika terjadi interrupt, LED akan menyala bergeser mulai dari LED bit 0 hingga bit 7 lalu kembali lagi ke LED bit 0.
 - Masing-masing interrupt hanya akan menampilkan satu nilai (interrupt pertama menampilkan '00000000b', interrupt kedua menampilkan '00000001b', dst)
 - gunakan perintah RL atau RR untuk menggeser tampilan LED, dan perintah semacam DJNZ atau CJNE sebagai pembanding untuk memperlama timer (seperti pada contoh progra percobaan).